NASA CR-166,007

NASA Contractor Report 166007

NASA-CR-166007
19820019246

OCULOMETER FOCUS AND MIRROR CONTROL

Warren J. Guy

LAFAYETTE COLLEGE
Easton, Pennsylvania 18042

Grant NAG1-45
October 1982

National Aeronautics and
Space Administration

Langley Research Center
Hampton, Virginia 23665

CONTENTS

FIGURES

OCULOMETER FOCUS

AND

MIRROR CONTROL

by

Warren J. Guy

1. Introduction

Since 1970 the Flight Dynamics and Control Division at the National Aeronautics and Space Administrations' Langley Research Center, Hampton, Va., has been engaged in a research program employing a remote sensing oculometer. The purpose of this program has been to gather and analyze dynamic data collected about pilot eye behavior. Ultimately it is hoped that this research will lead to strategies for optimal aircraft instrument panel design and better pilot training (1,2,3).

The heart of this research is built around the oculometer, a nonintrusive eye tracking instrument. The unit employs a safe infrared (IR) beam which is reflected off the retina and cornea to produce an image picked up by an IR sensitive video system. The picture is digitized and pertinent information processed by a mini-computer (4). The "look point" is the primary output of the system; however, other functions must also be performed by the computer.

As the subject's eye and/or head moves, it is necessary to maintain the eye image nearly centered in the TV field. This is accomplished with two mirrors orthogonally oriented which direct the IR beam to the subject and receive the reflected eye image. The mirrors, driven by galvanometer motors, are controlled by the minicomputer. The computer generates a control signal based on the motion of the eye as detected by the video signal. Often, due to sudden voluntary motion or involuntary motion caused by turbulence, the eye image is lost. At this point the tracking system must go into a lengthy and not always successful searching routine to recapture the eye image. During this search period all the data is lost, and depending upon conditions, well over 10% of the experiment time could be taken by searching.

In the normal computational process, the computer has 1/60th second between fields to find the look point and output the necessary control signals to the system peripheral equipment like the mirrors. The program requires a great deal of time (usually greater than 1/60th second) to perform its tasks. Due to the structure of the software, this means that alternate fields are lost. When this occurs, it is possible to lose track and thus be forced into a search mode.

The system as it is currently configured requires the constant attention of a highly skilled operator. It is his function to accelerate the search process, keep the optical system focused, determine the record linearization data for each experiment and maintain the equipment. Thus, work has been undertaken to speed up the computational process, miniaturize the oculometer system, unburden the main computer and reduce the operator skill level and intensity. To this end, this report discusses methods to decrease the required operator attention through automatic focusing and tracking. An important by-product of this effort is that the subject's range is constantly updated; thus providing the computer with data so that a more accurate look point can be calculated.

2. FOCUS CONTROL

The distance to the subject from the reference plane on which the look point is desired is determined in the NOVA computer by processing the voltage off the focus command (FC) potentiometer. This pot is manually adjusted by the operator as the subject moves his head position. While this chore is not overly burdensome, inaccurate focusing can lead to a fuzzy image, the possibility of losing track, and incorrect look point calculation.

The inaccurate look point can be caused by improper focus and the subsequent computational error produced in the corneal high light - pupil relationship. It can also be induced by an inaccurate measurement range; this latter type error will be considered. Figure 1 shows a y look point error, Δy. As can be seen from the figure, if the real range is R and there is a negative error, ΔR, the look point will be calculated higher by the amount

$$\Delta y = \Delta R \sin \theta$$

For a nominal range error of 2 inches and $\theta = 10^{\circ}$, the look point error could be 0.35 inches.



FIGURE 1. Range error induced look point offset.

Thus an automatic focusing mechanism for the oculometer will reduce these errors because more accurate range data will be available to the computer. Also, a sharp image will be maintained at all times.

To accomplish the automatic focus control, a simple position feedback control system is used, figure 2. An ultrasonic range detector and microprocessor develop the focus command. This signal (along with a potentiometer position feedback voltage) drives a standard servomechanism error amplifier. A power amplifier controls the small dc servomotor.

In the next section a systems overview of both the hardware and the software will be given. Then the focus table development will be described. Finally a detailed description of the electronics and microprocessor program will be presented.

2.1 SYSTEM OVERVIEW

<u>HARDWARE</u>: Figure 3. The range detector employs a commercially available ultrasonic ranging system (5), which consists of a capacitive microphone and a transceiver board. The unit must be supplied with a 6V, 3A (minimum) power supply and an external circuit to control the repetition rate of the transmit/receive cycle. The unit produces two output



FIGURE 2. Automatic Focus System basic structure.



FIGURE 3. Oculometer automatic focus system block diagram

pulses on separate lines, one when the transmitter is turned on and the other when an echo is received. The rate control must allow sufficient time for the ultrasonic pulse to travel to and return from the target. The specifications limit the time between transmit pulses to 40 msec.; the prototype automatic focus system is designed for a 100 msec. cycle time.

An interface board buffers the ranging system and control circuits. The transmit and echo pulses are next passed through the pulse shaper where they are cleaned up and gated to prevent false echos which would interrupt the microprocessor. Using an on-board timer, the microcomputer measures the time to the target, filters the data, generates a range count for display purposes and outputs a digital focus command (FC). The digital command is then converted to analog form.

The error amplifier can be switched between a manual and automatic focus control signal. In the manual mode, a simple potentiometer changes the focus. This is the way focus is currently adjusted. The FC command (manual or automatic) is compared with the lens position via a feedback potentiometer and a signal developed to drive a power amplifier. The final element in the system is the mechanical parts; motor, gears, and load.

When the system is out of focus, an indicator light is turned on.

SOFTWARE: Figure 4 shows the flow chart for the microcomputer program. A manual reset starts the program by clearing the appropriate working registers. The on-board timer is then cleared (START) and the subroutines SHOW and DISPLAY called. These two subroutines decode the binary range count (the round-trip time to the target) to a 3 digit BCD Code which is cycled through hundreds, tens, and units; the display is held on until the transmit pulse is sent. At this time the timer is activated (TIME) and the interrupt enabled. While waiting for the echo (typically 4 to 6 msec), the previous binary range count is converted to a decimal format by DISPLAY. The program continues its wait by returning to the SHOW and DISPLAY subroutines.

Upon the arrival of the echo, the microcomputer is interrupted (STOP), the timer stopped and checked. If the count is too large (greater than 90_{10}) the target is out of range and the maximum distance (90_{10}) is substituted for the range count.

The range count is now filtered by a simple digital filter algorithm (FILTER):

$$\text{OUT(I)} = \text{OUT(I-1)} + \frac{\text{COUNT(I)} - \text{OUT(I-1)}}{4}$$

where

COUNT (I) = the range count (typically 30_{10} - 90_{10})

COUNT (I) = the new filter output

OUT (I-1) - the previous filter output



FIGURE 4. Oculometer automatic focus flowchart

Finally OUT(I) is used to look up in a focus table the actual binary focus command (FC). This information is next latched into PORT 1. The program then cycles back to START where the whole process is repeated.

If the timer should overflow due to a malfunction or the lack of a target, the system ignores the range data and returns to START.

2.2 FOCUS COMMAND TABLE

OPTICAL SYSTEM: The optical system to be controlled is shown in figure 5. Here the ground glass plate, built into the demonstration model, would be the active surface of the TV camera in the oculometer. Lens F2 is mounted in a movable lens carriage. It is the distance L2 that will be controlled by the servomechanism. F1 is a fixed lens located a distance L1 from the plate. The subject is at a range R from the optical system.

The equation governing the relationship between the system variables is stated below:

$$F1*R = (R - F1)*(L1 - L2 + (F2*L2/(F2 + L2)))$$

where F1 = focal length of lens 1 (nominally 95mm)

F2 = focal length of lens 2 (nominally -29mm)

R = range to subject (nominally 600 to 1200mm)

L1 = fixed distance - lens 1 to plate (nominally 160mm)

L2 = variable distance - lens 2 to plate (nominally 60mm)

Rearranging the equation to find L2 in terms of range R gives the nonlinear relationship:

$$L2^2 + (F1*/(R-F1)-L1)*L2 + (F1*R/(R-F1) - L1)*F2 = 0$$

Because the computation of L2, with range as a variable, would require a fairly complex effort for the microprocessor and a large block of time, it was decided to use a look-up table to relate the two.



FIGURE 5. Optical system for demonstration unit.

RANGE-TIME RELATIONSHIP: The range measurement used in the focus command table comes from the microprocessor's on-board timer. It actually measures the time between the start of the transmitter and the received echo. For the system in question the timer counts pulses at the rate of one every 80 usec. (6) from an internal clock. For sound traveling at 13548 inches/sec., this gives a theoretical range/count relationship of:

$$\text{RANGE(inches)} = 0.54 * \text{COUNT}$$

Actual measurement of the system produced a slightly modified equation:

$$\text{RANGE(inches)} = -0.36 + 0.55 * \text{COUNT}$$

The offset error (-0.36 inches) may be due to the uncertainties in the start of the ultrasonic transmission and the timer or from inaccurate range measurements.

Figure 6 shows this experimentally obtained relationship.

FOCUS TABLE: Refer to figure 7. To obtain the final range/focus command (R/FC) table, a theoretical curve was first derived. It is based on a measured maximum of $\pm$ 7 volts output from the feedback potentiometer. The resulting curve is given in figure 7. This curve is very sensitive

FIGURE 6

RANGE/COUNT RELATIONSHIP

COUNT = 0.65 +1.82 RANGE

RANGE = -0.36 +0.55 COUNT





FIGURE 7
Voltage for focus command (DAC output, manual, and theoretical) vs. Range.

to small errors in L1, F1, and F2. Thus, when the theoretical focus table was loaded into the EPROM (8748), it was found that a good focus was not achieved over the entire range of interest.

A qualitative measurement was then made by focusing the optical system by eye and measuring the feedback potentiometer output. After repeating this experiment several times, a band of acceptable focus points for each range was obtained. This is indicated by the two solid curves in figure 7.

Values within these curves were then chosen for the focus table, converted to the appropriate DAC input and loaded into the EPROM. After several minor adjustments, this method produced a focus command which gave good results over the entire range. The final points entered in the focus table are the dark circles in figure 7.

2.3 FOCUS TABLE ADJUSTMENTS

The focus commands stored in the focus table (page 03 of the 8748 EPROM), are dependent on the particular mechanical relationship between the ultrasonic microphone, which detects the range, and the optical system, which focuses on the subject. To allow for differences in this relationship, five possible adjustments exist.

1. For each situation reprogram the 8748. This would entail a lengthy initial set-up procedure in which the focus voltages are measured, converted to a digital command and "burned" into the EPROM. Once done it would not have to be repeated for a particular installation.

2. Mechanically adjust the relationship (fine tune) between the microphone and the lens for the best focus over the ranges anticipated. At extremes, there will be focus errors.

3. Adjust the DAC offset potentiometer (to be discussed when the electronics are described). This shifts the focus curve, figure 7, up and down. Again this adjustment may be satisfactory over limited ranges, but will not be good at the extremes.

4. Adjust the focus table entry point by adding or subtracting a fixed amount; i.e., increment/decrement the range count. This moves the focus curve left or right and at extreme ranges again may result in excessive errors. A combination of 3 or 4 may give better results. Option 3 is a simple screwdriver adjustment. Option 4 requires only a simple program change, but will necessitate "burning" a new program into the EPROM for each adjustment. Like 1, once done it need not be repeated.

5. Store the focus table in RAM. The table would be entered at start-up for each run. It would therefore require a one-time initialization procedure. Also it would mean a significant design change of the electronics to allow proper interfacing with and data transfer from the mainframe computer.

Procedure 5 gives the greatest flexibility. Number 1 is the next best permanent option. 3 and 4 combined require the same effort as 1. All things considered, procedure 2 appears to be the simplest method to adjust the system for different experimental set-ups. This method was used during the design stages.

2.4 ELECTRONIC CIRCUIT

The front end of the automatic focus control unit consists of a commercial ranging system, figure 8. This system emits a 60, 57, 53, and 50 kHz sequential ultrasonic signal from a capacitive microphone. The transceiver board, powered by a user supplied 6V, 3A source, provides both the "start of transmit" signal and the "echo received" signal. Details of the operation can be found in reference 5.

The transceiver is controlled by a multivibrator built from a 74C14 Schmidt inverter and the 2N4401/2N2907 buffer.



FIGURE 8. Interface circuit for Polaroid Ultrasonic Range Measuring System. From Polaroid Handbook.

When the INITIATE TRANSMIT PULSE goes high, the transceiver board is activated and after a brief period (less than 1 msec.) the ultrasonic signal is started and maintained for about 1 msec. This action is detected in pin 16, bufferred by the 2N4401, and sent to the microprocessor.

The echo signal, when detected, causes a high pulse at pin 15. This signal is also bufferred by a 2N4401 and then used to drive the microprocessor.

Figure 9 shows the digital portion of the control board. The start and echo signals drive separate 74123's, dual monostables. The input signals initiate a 1 millisecond pulse from each package's first monostable; which, on its rising edge, activates the respective second monostable. A 10 usec. pulse is then available on pins 5 or 12. Pin 5 is a positive pulse (Q) and drives the microprocessor input TØ (used to start the on-board times). Pin 12 is a $\bar{Q}$ pulse which is used to interrupt, $\overline{INT}$, the microprocessor and stop the timer. The initial 1 msec monostable is used to block false start and echo signals that may appear on the lines due to echos from multiple targets or other noise sources.

The only manual control of the microporcessor is a reset button which, among other things, sets the program counter to 00_H. A 6 MH_z crystal controls the instruction cycle time and the on-board times. Two of the three available ports on the 8748 are used; one (Port 2) is tied to a



FIGURE 9. Part 1, microcomputer board. Pulse shaping (74123), microcomputer (8748), and DAC (1408 - 741).

display which will not be necessary in the final oculometer design, and the other (Port 1) outputs the digital focus command (FC).

FC is converted to analog form by an 8 bit DAC 1408. This DAC output is a 0 to -2 mA current which through a 741 op-amp is converted to an appropriate voltage as determined by the feedback resistor. The 10K potentiometer serves to provide bipolar operation from the DAC by biasing the op-amp so that with a 80_H input, the op-amp output is 0 V. This gives an offset binary code operation.

The analog signal (FC) is next used as a command signal to a standard position feedback circuit, figure 10. If the operator desires, the FC signal can be derived from a manually controlled potentiometer. In either case this signal is compared to the feedback signal generated by the potentiometer connected to the motor shaft. When there is an error signal, it is amplified by the two 741's and used to drive a complementary pair (NPN-2222 and PNP-2N2907). This pair in turn controls the current to the servomotor. Feedback from the top of the motor to the second 741 serves to reduce the dead zone created by the complementary transistors.

An LED indicator is also provided to inform the operator when the system is not responding to the FC command (manual or automatic). Two LM311's (comparators) set limits so that an error greater than $\pm$ 0.1 volts will turn on the LED.



FIGURE 10. Electronic controller and focus error indicator circuits.

The motor, gears, lens carriage, and mount were provided by the Flight Dynamics and Control Division, Langley Research Center, Hampton, VA.

The final circuit, figure 11, consists of a 74C48, BCD to 7 segment driver, three 7 segment displays and three 2N2222's used as switches. Port 2's lower 4 bits provide the BCD code while bits 4, 5, and 6 respectively turn on the units, tens, and hundreds display digits.

The final figure of this section, figure 12, shows the cable connections for the power supply and servo unit.

2.5 SOFTWARE

The heart of the automatic focus control unit is 8747 microcomputer on a chip. This unit has a 1K EPROM program memory, 64 byte data RAM, one on-board timer, and 27 pin I/O arranged as three 8 bit ports and three test pins.

In establishing the memory map, figure 13, the RAM is divided into two memory banks, RBØ and RB1. RBØ is used to store the range count (COUNT) from the timer, calculate the digitally filtered output (OUT), and store the previous output. RB1 is used to convert the binary to decimal and then to BCD and finally output the 3 digit display control signal.

The 1K EPROM contains all the permanent software(8). In



FIGURE 11. Display circuit. Input from Port 2, 8748 via flat 16 conductor cable.



MALE PLUG
POWER CABLE CONNECTION



MALE PLUG
SERVO CABLE CONNECTION

FIGURE 12. POwer cable and servo cable pin connection

MEMORY MAP



FIGURE 13.
Oculometer automatic focus memory map.

page ØØ is the main program. Figure 13 shows where the subroutines are located. This map can be compared with the flow chart in figure 4. Page Ø3 contains the focus table which is used to generate the proper focus command, given the range count (OUT(1)).

The digital filter uses a simple algorithm

$$\mathrm{OUT(I)} = (\mathrm{OUT(I-1)} + 0.25\ (\mathrm{COUNT\ (I)} - \mathrm{OUT(I-1)})$$

The 0.25 multiplier was used because of the ease with which it can be implemented; i.e., double shift right. It was determined after experimentally trying other 2^{-n} factors.

In the focus table, ranges from about 19" to 50" are allowed. Closer than 19" the lens carriage will have reached its limit. Ranges greater than 50" produce a count greater than 90 and are thus rejected. In this case the lens carriage is near its other limit and will remain so until the target range is less than 50".

The remaining pages of this section contain the program stored in the 8748. The format gives the line number in column 1, the memory address in column 2, either the hex machine code or data in column 3, column 4 is a label, the mnomic is in column 5 and lastly in column 6 are comments.

The details of the program are explained by the contents.

```
RANGE/FOCUS PROGRAM FOR OCULOMETER

AUTOMATIC FOCUS USING INTEL 8748

1-18-82, WJG

FORMAT-ADDRESS/DATA OR INSTRUCTION/LABEL/MNEMONIC/COMMENT


*********** INITIALIZE/START/TIME ***********

LOCATIONS 00-08 USED FOR RESET, INTERRUPT, AND TIMER OVER-
FLOW VECTORS.

00 04 RESET      JMP            JUMP TO INITIALIZE - CLEAR
01 10            INITIALIZE     REGISTERS.
02 00
03 04 INTER      JMP            JUMP TO INTERRUPT SERVICE -
04 30            STOP           STOP TIMER.
05 00
06 00
07 04 TOVER      JMP            TIMER OVERFLOW-OUT OF RANGE
08 20            START          BEGIN AGAIN.



10 27 INITIAL    CLR A          CLEAR REGISTERS
11 AA            MOV R2,A       R2=COUNT INTEGER
12 AB            MOV R3,A       R3=COUNT DECIMAL
13 AC            MOV R4,A       R4=OUT INTEGER
14 AD            MOV R5,A       R5=OUT DECIMAL
15 AE            MOV R6,A       R6=(COUNT(I)-OUT(I-1)) INTEGER
16 AF            MOV R7,A       R7=(COUNT(I)-OUT(I-1)) DECIMAL
17 04            JMP            BEGIN
18 20            START
```

```
20 27 START    CLR A          CLEAR ACCUMULATOR
21 62          MOV T,A        SET TIMER TO ZERO
22 04          JMP            JUMP TO DISPLAY OUTPUT WHILE
23 A0          SHOW           WAITING FOR TRANSMIT START
24 55 TIME     STRT T         START TIMER
25 05          EN I           ENABLE INTERRUPT TO STOP TIMER
26 C5          SEL RB0        ON ECHO RETURN
27 04          JMP            JUMP TO CALCULATE NEW RANGE
28 80          DISPLAY        COUNT DISPLAY



30 65 STOP     STOP TCNT      STOP TIMER
31 15          DIS I          PREVENT FALSE ECHOS
32 C5          SEL RB0
33 42          MOV A,T
34 AA          MOV R2,A       R2=COUNT(I)
35 03          ADD A,#26      CHECK COUNT LIMIT > 90
36 26
37 F2          JB7            IF COUNT > 90 SET LIMIT
38 3B          LIMIT
39 04          JMP            IF NOT, FILTER
3A 3E          FILTER
3B 23 LIMIT    MOV A,#5A      SET COUNT =90
3C 5A
3D AA          MOV R2,A
3E FB FILTER   MOV A,R3       R3=DECIMAL COUNT=00H
3F 37          CPL A
40 6D          ADD A,R5       R5=DECIMAL OUT(I-1)
41 37          CPL A
42 AF          MOV R7,A       R7=DECIMAL (COUNT(I)-OUT(1-1))
43 FA          MOV A,R2       R2=INTEGER COUNT(1)
44 37          CPL A
45 7C          ADDC A,R4      R4=INTEGER OUT(1-1)
46 37          CPL A
47 AE          MOV R6,A       R6=INTEGER (COUNT(1)-OUT(I-1))
48 85          CLR F0
49 B8          MOV R0,#02
4A 02
4B E6          JNC
4C 4E          SHIFT
4D 95          CPL F0         SET F0 WHEN (COUNT(1)-OUT(I-1))<(
```

```
4E 97 SHIFT     CLR C           DIVIDE (COUNT(I)-OUT(I-1)) BY
4F FE           MOV A,R6        4 WITH 2 RIGHT SHIFTS
50 67           RRA C
51 AE           MOV R6,A
52 FF           MOV A,R7
53 67           RRA C
54 AF           MOV R7,A
55 E8           DJNZ R0
56 4E           SHIFT
57 B6           JF0             ADJUST INTEGER PART IF DIFFERENCE
58 5B           SUB             NEGATIVE
59 04           JMP             OTHERWISE ADD
5A 5F           ADD
5B FE SUB       MOV A,R6        INTEGER ADJUST
5C 43           ORL A,#C0
5D C0
5E AE           MOV R6,A        COMPLETE FILTER ADDITION
5F FD ADD       MOV ,R5         OUT(I)=OUT(I-1)+0.25*(COUNT(I)-
60 6F           ADD A,R7               OUT(I-1))
61 AD           MOV R5,A
62 FC           MOV A,R4        STANDARD FIRST ORDER FILTER
63 7E           ADDC A,R6       WITH TF=TS
64 AC           MOV R4,A
65 E3           MOVP3 A,@A      FIND FOCUS COMMAND
66 39           OUTL P1,A       OUTPUT COMMAND
67 04           JMP             DO IT AGAIN
68 20           START







80 FA DISPLAY   MOV A,R2         PUT COUNT(I) INTO ACCUMULATOR
81 D5           SEL RB1          WORK IN REGISTER BANK 1
82 A8           MOV R0,0         R0=REMAINDER STORAGE
83 BF           MOV R7,#40       R7=HUNDREDS OUTPUT/CONTROL
84 40                               40H=CONTROL
85 BE           MOV R6,#20       R6=TENS OUTPUT/CONTROL
86 20                               20H=CONTROL
87 BD           MOV R5,#10       R5=UNITS OUTPUT/CONTROL
88 10                               10H=CONTROL
89 BC           MOV R4,#9C       R4=HUNDREDS SUBTRACTOR
8A 9C
8B BB           MOV R3,#F6       R3=TENS SUBTRACTOR
8C F6
```

```
8D 6C HUNDRED   ADD A,R4        SUBTRACT 100 FROM COUNT, IN-
8E E6           JNC             CREMENT HUNDREDS COUNTER IF
8F 94           TENS            SUBTRACTION SUCCSSFUL.  CON-
90 AB           MOV R0,A        TINUE.  OTHERWISE GO TO TENS.
91 1F           INC R7
92 04           JMP
93 8D           HUNDREDS
94 F8 TENS      MOV A,R0        SAME AS BEFORE EXCEPT FOR
95 6B           ADD A,R3        TENS.
96 E6           JNC
97 9C           UNITS
98 AB           MOV R0,A
99 1E           INC R6
9A 04           JMP
9B 94           TENS
9C FD UNITS     MOV A,R5        REMAINDER IS UNITS, ADD TO
9D 68           ADD A,R0        R5.
9E AD           MOV R5,A
9F 00           NOP




A0 D5 SHOW      SEL RB1         SELECT REGISTER BANK 1 WITH
A1 FF           MOV A,R7        DISPLAY DATA AND CONTROL
A2 3A           OUTL P2,A       OUTPUT HUNDREDS
A3 14           CALL
A4 B0           DELAY
A5 FE           MOV A,R6
A6 3A           OUTL P2,A       OUTPUT TENS
A7 14           CALL
A8 B0           DELAY
A9 FD           MOV A,R5
AA 3A           OUTL P2,A       OUTPUT UNITS
AB 14           CALL
AC B0           DELAY
AD 04           JMP             DO IT AGAIN
AE A0           SHOW





B0 27 DELAY     CLR A          DISPLAY DIGIT
B1 17 X         INC A
B2 36           JT0            IF T0 HIGH, START TIMER
```

```
B3 24          TIME
B4 96          JNZ           OTHERWISE, CONTINUE THE
B5 B1          X             DISPLAY
B6 93          RETR
```

```
300 00 RANGE  RANGE/COUNT DATA  FOCUS CONTROL DATA AT COUNT
                                ADDRESS, PAGE 3 OF MEMORY.

321 00
322 1D        18.5"/34
323 3A        19.0"/35
324 3F
325 46        20.0"/37
326 4A
327 4F        21.0"/39
328 54
329 5A        22.0"/41
32A 5C
32B 62        23.0"/43
32C 6A        24.0"/44
32D 6C
32E 70        25.0"/46
32F 74
330 78        26.0"/48
331 79
332 7B        27.0"/50
333 7D
334 80        28.0"/52
335 83        29.0"/53
336 86
337 89        30.0"/55
338 8A
339 8C        31.0"/57
33A 8D
33B 8F        32.0"/59
33C 91
33D 93        33.0"/61
33E 95
33F 98        34.0"/63
340 9A        35.0"/64
341 9B
342 9C        36.0"/66
343 9E
344 A0        37.0"/68
345 A1
346 A2        38.0"/70
347 A3
348 A4        39.0"/72
349 A5        40.0"/73
34A A6
34B A7        41.0"/75
34C A8
34D A9        42.0"/77
34E AA
34F AB        43.0"/79
350 AC
351 AD        44.0"/81
352 AE
353 AF        45.0"/83
354 B0
355 B1        46.0"/85
356 B2        47.0"/86
```

```
357 B3
358 B4          48.0"/88
359 B4
35A B5
35B B5
35C B6
35D B6
35E B7
35F B7
360 00


380 00
```

3. TRACKING CONTROL

One of the major improvements which could be made in the present oculometer system is a better search routine. This task now takes valuable time from data acquisition.

When the subject's eye image is lost, the software drops into a locate and verify process. To accomplish this, the mirror control unit is directed to search in a small area by scanning in the X direction, the incrementing in Y and repeating the X scan. If the small search area is not fruitful, a larger area is searched. This process if repeated (X-Y scan) until an eye image with round pupil and corneal highlight is obtained. It is then verified and only then can valid look point data be collected. The search process is not always successful. When this happens, large blocks of time are used in repetitously repeating the same X-Y scan pattern.

If the eye and/or head moves slightly, so that the image stays in the picture, minor adjustment in the mirror position are made by the software. These attempt to keep the pupil center in the TV picture. This segment of the oculometer's operation works well.

The purpose of the unit developed for head tracking is to supplement the oculometer when a search is necessary. It does not replace the fine adjustments made when the eye is centered. The unit will basically tell the computer where the head is located and direct the mirrors to that location by using X-Y ultrasonic ranging systems. By employing a known offset, the eye can be found. The geometry of the system's operation is shown in figure 14.

3.1 SYSTEM OVERVIEW

HARDWARE: Refer to figures 14 and 15. The overall system operation originates at the rate controller which alternately switches between the X and Y ultrasonic systems. The ultrasonic system is the same (7) as used in the automatic focusing unit. Both the X and Y ultrasonic boards are buffered to give compatible signals to the wave-shaping circuit and microprocessor.

The only difference between the focusing unit's ultrasonic operation and the head tracking system is the X-Y rate controller. Its signal, a 200 msec. square wave, is fed into the microprocessor as well as the ultrasonic units. When high, a Y distance measurement is being made. A low level indicates that the X direction is being measured. Thus, once every 200 msec., the X transmitter



FIGURE 14. Tracking system geometry.



FIGURE 15. Automatic Tracking System
block diagram.

is turned on for about 1 msec.; the receiver then waits for an echo for 100 msec. The X unit is then turned off and the same operation repeated for Y.

Three signals, START of transmission, ECHO received, and X-y Direction indicator are fed into the microcomputer. This information is processed and outputted on Port 1 for X and Port 2 for Y. The 8 bit output signals are converted to analog form by two DAC's.

The analog signal is now fed to a power amplifier which converts the voltage input to a current output. Where in turn it drives two orthogonal galvanometer motors. The mirrors, motors, and their housing were also supplied by NASA. This last link in the system directs the IR beam to the target.

There is a provision to manually control the mirrors with a "joy stick." Voltage for the joy stick is from a $\pm$1.25V regulated supply. Also, the system can be manually reset.

SOFTWARE: Figure 16 gives the flow chart for the software programmed into the EPROM memory of the microcomputer.

After a manual RESET, the program jumps to INITIALIZE where registers are cleared and the X-Y offset loaded. Note that for different mirror-microphone relationships, these values would have to be changed, see figure 14.


MANUAL RESET
INITIALIZE
CLEAR REGISTERS
LOAD X = 32H IN R34
Y = 23H IN R35
TIMER OVERFLOW
A
START
LOAD RANGE IN R7
(SET FOR 36" = 42H)
CLEAR TIMER
TØ
LOW
HIGH
START TIMER
T1
HIGH
LOW
Y FLAG
RØ=33H
X FLAG
RØ=32H
INTERRUPT (ECHO)
WAIT
WAIT FOR INTERRUPT
STOP
STOP TIMER
STORE x/y COUNT
DIFF
CALCULATE (X-x) or (Y-y)
SET FØ IF NEGATIVE



FIGURE 16. Automatic Tracking System flow chart.

The particular offset presently loaded is for 19" in the Y direction and 27" in the X direction.

The program now jumps to START where the target range is next stored. This is also fixed in the demonstration unit at 36", but could be constantly updated by the automatic focusing unit if the same microcomputer controlled all functions (see Conclusions). The Timer is cleared and the system put into an idle mode until TØ goes high. TØ corresponds to the START of the ultrasonic transmission. The Timer is started and, while waiting for the ECHO, and X-Y flag is set to indicate which direction is being measured. The flag is controlled by T1 which is reading the high-low status of the ultrasonic rate controller.

When the INTERrupt goes low, meaning that an ECHO has been received, the Timer is STOPped and the distance count stored.

The DIFFerence between the offset X/Y and the measured distance x/y is next calculated. A flag FØ is set for a negative result.

To find the angle, θ, which controls the mirrors movement, the DIVIDE routine is used (9). The approximation

$$\theta \approx \sin\theta \approx \frac{\text{distance}}{\text{hypotenuse}}$$

is used in this algorithm since the angles will be less than $\pm\ 14^{\circ}$. For the particular application

$$\theta_x = (X - x)/\text{Range}$$

or

$$\theta_y = (Y - y)/\text{Range}$$

The output of this computation is in a sign(FØ) -magnitude format. To correctly operate the digital to analog converter, an offset binary is required. The transformation between the two is provided by CONVERT (and a potentiometer setting).

The program next drops down to OUTPUT which latches the command θ into Ports 1 or 2 if the measurement was X or Y respectively. The program then branches back to START and recycles through the entire operation.

Should there be a Timer overflow due to the lack of a target (no ECHO), the program will also return to the START routine to continue its cycle.

3.2 ELECTRONIC CIRCUIT

The electronic circuit diagrams for the head tracking system are shown in figures 17 through 22.



FIGURE 17. X ultrasonic board with buffer and rate controller.

Figure 17 is the X ultrasonic unit. It is mounted so as to measure target motion in the X direction only. It consists of an ultrasonic ranging system board with a capacitive microphone and a buffer circuit as shown. This is a speed up circuit which allows pulsing at a 50 msec rate. Note that the START and ECHO signals swing between 0 to 5 volts. These are fed to the wave shaping network. Also included in the X ultrasonic unit is the multivibrator rate controller for alternating X and Y measurements. This made from a 74C14 and a simple RC timer. The X_c signal likewise controls the Y unit and is used as an input to the microprocessor.

Figure 18 is the ultrasonic Y unit and is identical to the X unit except that no rate controller is present.

In figure 19 the START and ECHO signals from X and Y are fed into NOR gates which output a pulse when transmission starts and an echo is received. The pulses are used to trigger the 74123's dual monostables. The first monostables in each 74123 is set to about 10 msec. The second, driven by the first, gives only a 20 usec pulse which drives the microprocessor. The first monostable time is long in order to isolate the microprocessor from noise and false echos (from multiple targets). The START signal triggers the TØ pin and the ECHO signal triggers the $\overline{\text{INT}}$ pin. A 6MHz crystal controls the speed of operation of the processor and the timer.



FIGURE 18. Y ultrasonic board with buffer.

FIGURE 19. Wave shaping circuit (74123) and microcomputer (8748)



The RESET switch discharges the capacitor through a 1K to force the reset pin low. The T1 input is from Y_c which signals the processor as to which direction, X or Y, is being measured. The output Port 1 is to the X mirror control, while Port 2, divided into two 4 bit nibbles, controls the Y mirror.

In figure 20 the outputs from the 8748 drive two 1408 DAC's that are set up for bipolar operation. The 8 bit input is converted to a 0 to -2mA output. This is combined with currents from a 5 volt source through a 22K and 10K pot. One-half a 4558 op-amp is used for each direction to convert the current to an output voltage. The 22K-10K pot parallel combination allows adjustment to the reference position in a manner similar to that provided by the X-Y offset value.

Figure 21 shows the regulated voltage source for the manual joy stick control. It is composed of an LM317 (positive voltage) and LM337 (negative voltage) adjustable voltage regulators. The input to the circuit is $\pm$15V, its output $\pm$1.25V. A switch allows selection between manual and automatic control. The X_{out}-Y_{out} is connected to the driver amplifiers.

Finally figure 22 gives the pin connections for the interconnecting cables. The first, a 10 conductor ribbon cable, is for the X-Y START/ECHO signals, the direction signal, and a 5V source. The other cable carries the 6V for the ultrasonic boards.



FIGURE 20. DAC output (1408 - 4558)



FIGURE 21. Regulated supply for manual joy stick control.

8 PIN PLUG

1	BLACK	5V gd.
2	WHITE	+5V
3	GRAY	Y ECHO
4	PURPLE	Y START
5	BLUE	x/y MEASUREMENT
6	BROWN	X ECHO
7	RED	X START

4 PIN PLUG

2	BLACK	6V gd.
4	BLACK	+6V

FIGURE 22. Cable connections for Automatic Tracking System.

3.3 SOFTWARE

The program memory as shown in figure 23 is divided into various routines which correspond to those shown in the flow chart, figure 16. The entire program takes up less than 1 page of memory and can even be condensed further if all the routines were compacted one directly behind the other.

The on-board registers are used as indicated in figure 23. In register bank Ø, RBØ, register RØ is the flag which points to X or Y data as determined by the Y_c signal. R1 points to the divisor for the divide routine. This is R7 in the demonstration unit. R7 contains the range information. R1 also points to the X or Y offset (R34 or R35) during the DIFF routine. R2 is the dividend and R3 a counter for the DIVIDE routine. R3 is also a temporary storage location during OUTPUT.

R32 and R33 store the most recent X and y measured distances respectively. R34 and R35, the X and Y fixed offset, must be altered if the relationship between the mirrors and the ultrasonic microphones is changed.

The remainder of this section is the actual program stored in the EPROM of the 8748. Comment statements describe its operation. The format is the same for the automatic focus control unit Software discussion.

MEMORY MAP

PROGRAM MEMORY

PAGE Ø

ØØH	VECTORS FOR RESET, INTERRUPT TIMER OVERFLOW
1ØH	INITIALIZE
3ØH	START
5ØH	WAIT
6ØH	STOP
7ØH	DIFF
9ØH	DIVIDE
CØH	CONVERT
DØH	OUTPUT

REGISTERS

RBØ

ØH	x/y POINTER
1H	DIVISOR POINTER/X,Y OFFSET
2H	DIVIDEND
3H	COUNTER/OUTPUT STORAGE
7H	RANGE COUNT
32H	x COUNT
33H	y COUNT
34H	X OFFSET
35H	Y OFFSET

FIGURE 23. Automatic Tracking System memory map.

```
MIRROR CONTROL PROGRAM FOR OCULOMETER

AUTOMATIC HEAD TRACKING USING INTEL 8748

4-30-82, WJG

FORMAT-ADDRESS/DATA OR INSTRUCTION/LABEL/MNEMONIC/COMMENT


**********************************************************

LOCATION 00-08 USED FOR RESET, INTERRUPT, AND TIMER OVERFLOW
VECTORS.



00 04 RESET   JMP             JUMP TO INITIALIZE
01 10           INITIALIZE
02 00
03 04 INTER   JMP             JUMP TO INTERRUPT SERVICE
04 60           STOP          STOP TIMER
05 00
06 00
07 04 TOVER   JMP             JUMP TO START - TIMER OVERFLOW
08 30           START         OUT OF RANGE.




10 27 INIT    CLR A           CLEAR ACCUMULATOR
11 A9         MOV R1,A        CLEAR REGISTERS R1 TO R7
12 AA         MOV R2,A
13 AB         MOV R3,A
14 AC         MOV R4,A
15 AD         MOV R5,A
16 AE         MOV R6,A
17 AF         MOV R7,A
18 B8         MOV R0,#34      R0 IS POINTER TO R34
19 34
1A B0         MOV @R0,#32     32 TO R34 (X OFFSET=27")
1B 32
1C 18         INC R0          R0=35
1D B0         MOV @R0,#23     23 TO R35 (Y OFFSET=19")
1E 23
1F 04         JMP
20 30           START
```

```
30 BF START   MOV R7,#42    R7=RANGE COUNT, SET AT 36"
31 42
32 27 TIMSTRT CLR A         CLEAR ACCUMULATOR
33 62         MOV T,A       SET TIMER TO ZERO
34 26 IDLE    JNT0          IDLE UNTIL T0 HIGH
35 34
36 55         STRT T        START TIMER
37 00         NOP
38 56         JT1           SET X/Y FLAG
39 3E           YFLG
3A B8 XFLG    MOV R0,#32    X MEASUREMENT, R0=32
3B 32
3C 04         JMP           JUMP TO INTERRUPT WAIT
3D 50           WAIT
3E B8 YFLG    MOV R0,#33    Y MEASUREMENT, R0=33
3F 33
40 04         JMP
41 50           WAIT

50 86 WAIT    JNI           WAIT HERE FOR INTERRUPT
51 60           STOP
52 04         JMP
53 50           WAIT

60 65 STOP    STOP TCNT    STOP TIMER
61 00         NOP
62 00         NOP
63 42         MOV A,T      MOVE COUNT TO ACCUMULATOR
64 A0         MOV @R0,A    STORE X/Y COUNT IN R32/R33
65 04         JMP          JUMP TO DIFFERENCE
66 70           DIFF
```

```
70 F8 DIFF   MOV A,R0     MOVES X/Y POINTER TO ACCUMULATOR
71 17        INC A        POINTS TO X/Y OFFSET REGISTER
72 17        INC A
73 A9        MOV R1,A     OFFSET POINTER IN R1
74 F0        MOV A,@R0    ACCUMULATOR=X/Y
75 37        CPL A
76 17        INC A        ACCUMULATOR=-X/Y
77 61        ADD A,@R1    ACCUMULATOR=(X-X) OR (Y-Y)
78 F6        JC
79 80          PLUS
7A 37 MINUS  CPL A
7B 17        INC A        ACCUMULATOR=/(X-X)/ OR /(Y-Y)/
7C 85        CLR F0
7D 95        CPL F0       SET F0 FO R NEGATIVE RESULT
7E 04        JMP          GO TO DIVIDE
7F 90          DIVIDE
80 85 PLUS   CLR F0       CLEAR F0 FOR POSITIVE RESULT
81 04        JMP          GO TO DIVIDE
82 90          DIVIDE



                          16 BIT BY 8 BIT FIXED POINT UN-
                          SIGNED DIVIDION.
                          R1 POINTS TO DIVISOR REGISTER
                          MS BYTE IN ACCUMULATOR
                          LS BYTE IN R2
90 B9 DIVIDE MOV R1,#07
91 07
92 BB        MOV R3,#08
93 08
94 37        CPL A
95 61        ADD A,@R1
96 37        CPL A
97 F6        JC
98 9C          DIVIA
99 A7        CPL C
9A 04        JMP
9B B5          DIVIB
9C 61 DIVIA  ADD A,@R1
9D 97 DIVILP CLR C
9E 2A        XCH A,R2
9F F7        RLC A
A0 2A        XCH A,R2
A1 F7        RLC A
A2 E6        JNC
A3 A9          DIVIE
A4 37        CPL A
A5 61        ADD A,@R1
A6 37        CPL A
A7 04        JMP
A8 B1          DIVIC
```

```
A9 37 DIVIE   CPL A
AA 61         ADD A,@R1
AB 37         CPL A
AC E6         JNC
AD B1           DIVIC
AE 61         ADD A,@R1
AF 04         JMP
B0 B2           DIVID
B1 1A DIVIC   INC R2
B2 EB DIVID   DJNZ R3
B3 9D           DIVILP
B4 97         CLR C
B5 2A DIVIB   XCH A,R2
B6 04         JMP             RESULT IN ACCUMULATOR
B7 C0           CONVERT

C0 B6 CONVER JF0              CONVERSION FROM SIGN MAGNITUDE TO
C1 C7 POS       NEG           OFFSET BINARY. JUMP TO NEGATIVE.
C2 97         CLR C
C3 A7         CPL C           COMPLIMENNT SIGN BIT
C4 67         RRC A           MOVE OFFSET BINARY TO ACCUMULATOR
C5 04         JMP             GO TO OUTPUT
C6 D0           OUTPUT
C7 97 NEG     CLR C           CLEAR CARRY (SIGN BIT) FOR NEGATIVE
C8 37         CPL A
C9 17         INC A           CONVERT SIGN-MAGNITUDE TO OFFSET BINA
CA 67         RRC A           MOVE OFFSET BINARY TO ACCUMULATOR
CB 04         JMP
CC D0           OUTPUT

D0 2B OUTPT  XCH A,R3         STORE COMMAND IN R3
D1 F8        MOV A,R0
D2 12        JB0              JUMP IF B0 SET (R0=33) AND THUS Y
D3 D8          YP2
D4 2B XP1    XCH A,R3         OUTPUT FROM R3
D5 39        OUTL P1,A        OUTPUT X COMMAND TO P1
D6 04        JMP              GO BACK TO BEGIN
D7 32          TIMSTRT
D8 2B YP2    XCH A,R3         OUTPUT FROM R3
D9 3A        OUTL P2,A        OUTPUT Y COMMAND TO P2
DA 04        JMP              GO TO BEGIN
DB 32          TIMSTRT
```

4. CONCLUSIONS

Both projects, automatic focusing and head tracking, have proved the feasibility of such an approach to the improvement of the oculometer system. Major tasks still exist if the schemes herein reported are to be successfully incorporated into the Langley oculometer.

First the ultrasonic boards should be redesigned for this specific application. The present system produces audible "clicks" as it turns on, can operate only at limited rates, and has a narrow ultrasonic beam. These problems could be eliminated with a custom ultrasonic ranging system.

A scheme to bounce the ultrasonic beam off the mirrors and thus have it travel along the same path as the IR beam would help the automatic focusing unit. This would insure that the range, and hence the focus control, is directed at the eye.

It is possible to incorporate both focusing and tracking systems into a single unit. Thus one 8748 could control both operations, thereby reducing the complexity of the total.

There is enough memory to do this, but some major hardware modifications would be necessary.

Integration of these systems with the oculometer should be a paramount consideration. The range data and mirror control information can be advantageously used by the minicomputer. It may even be possible to do some preprocessing of these signals to further unburden the minicomputer.

Finally, a simple initialization routine should be developed so different sites can easily be set up. This could be incorporated through minicomputer control or by the manual provisions contained on the focusing and tracking units.

REFERENCES

(1) Middleton, D. B. et al, "Description of Flight Tests of an Oculometer," NASA Technical Note D-8419, June 1977.

(2) Spady, Amos A., Jr., "Airline Pilot Scan Patterns During Simulated ILS Approaches," NASA Technical Paper 1250, October 1978.

(3) Fulton, C. L. and R. L. Harris, Sr., "Error Analysis and Corrections to Pupil Diameter Measurements with Langley Research Center's Oculometer," NASA Technical Memorandum, May 1980.

(4) Waller, M. C. and M. A. Wise, "The Oculometer in Flight Management Research," paper, AIAA 14th Aerospace Science Meeting, Pasadena, CA, January 1975.

(5) "Ultrasonic Ranging System," Polaroid Corp., Cambridge, Mass., Jan. 1980.

(6) "MCS-48 Family of Single Chip Microcomputers User's Manual," Intel Corp., Santa Clara, Calif., August 1980.

(7) "8048 Family Applications Handbook," Intel Corp., Santa Clara, Calif., January 1980.

(8) "Prompt 48 Microcomputer User's Manual," Intel Corp., Santa Clara, Calif., 1978.

1. Report No. NASA CR-166007	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle Oculometer Focus and Mirror Control		5. Report Date October 1982
		6. Performing Organization Code
7. Author(s) Warren J. Guy		8. Performing Organization Report No.
		10. Work Unit No.
9. Performing Organization Name and Address Department of Electrical Engineering Lafayette College Easton, PA 18042		11. Contract or Grant No. NAG1-45
		13. Type of Report and Period Covered Contractor Report
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546		14. Sponsoring Agency Code

15. Supplementary Notes

Langley technical monitor: Patrick A. Gainer
Final Report (March 1, 1981 to April 30, 1982)

16. Abstract

In order to adapt the oculometer used at Langley Research Center to the specific eyetracking tasks being investigated, various studies are underway. This report describes one of these, an automatic focusing system designed around an ultrasonic range measurement. Besides maintaining the focus, subject distance is a by-product which could lighten the computational effort. The other system which is discussed is an automatic head tracking unit. It is intended to reduce the search time required when track is lost. An X-Y ultrasonic measurement is also made in this system to control the deflection mirrors.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Eye tracker Ultrasonic tracker Range measurements Automatic focusing	Unclassified - unlimited Subject Category 35

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price
Unclassified	Unclassified	68	A04

For sale by the National Technical Information Service, Springfield, Virginia 22161

End of Document